NF01 – Printemps 2018

Examen Final

Corrigé

1ère Partie : Tableaux *(5 points)*

Le triangle de Pascal est le triangle dont la ligne d'indice n (n = 0, 1, 2...) donne les coefficients binomiaux C_n^p pour p = 0, 1, 2..., n. (voir figure ci-dessous).



1								
1	1							
1	2	1						
1	3	3	1					
1	4	6	4	1				
1	5	10	10	5	1			
1	6	15	20	15	6	1		
1	7	21	35	35	21	7	1	
1	8	28	56	70	56	28	8	1

Pour avoir un terme de la ligne suivante, on prend le terme juste au-dessus, et on lui additionne celui qui est juste avant (0 s'il n'y a rien).

a. Définir une procédure `calculerTriangle(m,n)` qui remplit une matrice `m` avec un triangle de Pascal d'indice `n` construit suivant le principe décrit ci-dessus. L'indice `n` représente la dimension effective de la matrice avec `n` ≤ `MAX` et `MAX = 20`. La partie triangulaire supérieure doit être mise à 0.

b. Définir une procédure `afficherTriangle(m,n)` qui utilise la matrice `m` de dimension `n` pour afficher seulement la partie triangulaire inférieure (comme dans le tableau ci-dessus).

c. Ecrire une fonction `symetrique(t,n)` permettant de vérifier qu'un tableau `t` de `n` nombres entiers est constitué symétriquement de nombres identiques.

Exemple :

1									MAX
14	9	25	25	9	14				

d. Donner une expression permettant de vérifier que les éléments de la ligne `i` de la matrice du triangle de Pascal sont symétriques.

```
procedure calculerTriangle(var m:matrice, n:integer);
var n,i,j: integer;
begin
   for i:=1 to n do
      for j:=1 to n do
         m[i,j] :=0;
   for i:=1 to n do m[i,1] := 1;
   for i:=2 to n do
```

```
      for j:=2 to i do
         m[i,j] := m[i-1,j-1] + m[i-1,j];
end;

procedure afficherTriangle(m:matrice; n: integer);
var i,j: integer;
begin
   for i:=1 to n do
    begin
     for j:=1 to i do write(m[i,j],' ');
      writeln;
    end;
end;

function symetrique(t:tableau; n: integer): boolean;
var
   i:integer;
begin
   symetrique := true;
   for i:=1 to n div 2 do
     begin
      if t[i] <> t[n-i+1] then
         symetrique := false;
     end;
end;
```

Question 4 : symetrique(m[i],i)

{ ---------- pensez à changer de copie ---------- }

2ème Partie : Récursivité *(5 points)*

Un principe connu depuis fort longtemps permet d'effectuer la multiplication de deux entiers naturels a et b. Il est défini comme suit :

- si b vaut 0 alors : $a \times b$ est égal à 0
- si b est impair alors : $a \times b$ est égal à $a + a \times (b - 1)$
- si b est pair alors : $a \times b$ est égal à $(a + a) \times (b / 2)$

1. Ecrire une fonction récursive nommée *produit* qui applique ce principe pour multiplier deux nombres entiers a et b.
2. Faire une simulation avec a = 4, b = 3.

{ ---------- pensez à changer de copie ---------- }

3ème Partie : Fichiers *(5 points)*

On considère le programme incomplet suivant :

```
program MesPhotos;

const
    N = 100;

type
    ListTag = array [1..N] of string;
```

```
    Image = record
                nomFichier : string; //nom du fichier de la photo
                tags : listTags ; //tags pour commenter la photo
            end;
    FichierImages = file of Image;

var
   fApril, fRaphael : FichierImages;

begin
    assign(fRaphael, 'AlbumRaphael.txt');
    rewrite(fRaphael);

    assign(fApril, 'AlbumApril.txt');
    rewrite(fApril);
    …
    close(fRaphael);
    close(fApril);
end.
```

L'utilisateur peut donc associer à chaque photo d'un album des tags comme le nom des personnes se trouvant sur la photo (e.g. Leonardo, Raphaelo), l'événement associé à la photo (e.g. PizzaParty, SoireePic), etc.

1) Ecrire une procédure `saisie` en Pascal qui permette de saisir autant de photos que l'utilisateur le souhaite dans un fichier de type `FichierImages` passé en paramètre. La procédure demande à l'utilisateur le nombre de photos qu'il souhaite saisir. On veillera à ce que ce nombre ne dépasse pas `N`[1]. Pour chaque photo, la procédure demande : le nom du fichier de la photo, le nombre de tags et les tags associés à cette photo *(2 points)*.

   Par exemple, l'appel `Saisie(fRaphael)` pourrait permettre de saisir les informations suivantes dans le fichier :

   ```
   20180625_1.jpeg, tags : Leonardo, Michelangelo, PizzaParty,
   20180625_2.jpeg, tags : Michelangelo, PizzaParty,
   20180625_3.jpeg, tags : Leonardo, Donatello, PicParty, Shredder,
   20180625_4.jpeg, tags : Donatello, PizzaParty,
   ```

2) Ecrire une fonction `recherchePhoto` en Pascal qui demande à l'utilisateur le nom du fichier de la photo qu'il recherche et qui renvoie la position à laquelle se trouve cette photo dans le fichier `fImages` passé en paramètre, et -1 si la photo n'existe pas *(1 point)*.

   Par exemple, l'appel `i:= recherchePhoto(fRaphael)` afficherait :

   ```
   ***Quel fichier recherchez-vous :
   20180625_2.jpeg
   --- Nom trouvé en position : 1
   i sera égal à 1 dans ce cas
   ```

[1] Pensez à la façon de gérer par la suite de manière intelligente le nombre de tags

3) Ecrire une procédure `rechercheTag` en Pascal qui demande le tag recherché et qui affiche toutes les photos du fichier `fImages` passé en paramètre portant ce tag. La procédure affiche aussi le nombre total de photos possédant ce tag *(2 points)*.

Par exemple, l'appel `rechercheTag(fRaphael)` afficherait :

```
***Quel tag recherchez-vous :
PizzaParty
--- Tag trouvé dans photo 20180625_1.jpeg
--- Tag trouvé dans photo 20180625_2.jpeg
--- Tag trouvé dans photo 20180625_4.jpeg
Nombre de photos trouvées : 3
```

```
program GestiondeMesPhotos;
CONST
    N = 10;
type
   listTag = array [1..N] of string;
   image = record
      nomFichier : string;
      tags : listTag;
      end;
   fichierImages = file of Image;

var
  fImages : fichierImages;
  choix : integer;

function SaisiePhoto : image;
var
  nb, i : integer;
  photo : image;
begin
    write('-- Saisir nom du fichier : ');
    readln(photo.nomFichier);
    repeat
        writeln('   Combien de tags voulez vous saisir (nb < ', N, ')');
        readln(nb);
    until (nb >= 0) and (nb < N);
    for i := 1 to nb do
       begin
          write('        Tags ', i:2, ', Saisir nom du tag : ');
          readln(photo.tags[i]);
       end;
    photo.tags[i+1] := '';
    writeln;
    SaisiePhoto := photo;
end;

procedure Saisie (var f : FichierImages);
var
  nb, i: integer;
```

```
begin
    repeat
        writeln('Combien d images voulez vous saisir ');
        readln(nb);
    until (nb > 0);
    reset(f);
    for i := 1 to nb do
         write(f, SaisiePhoto);
end;

procedure Affichage (var f : FichierImages);
var
  i : integer;
  photo : image;
begin
    writeln('---- Affichage Photos ---');
    reset(f);
    while (not (eof(f))) do
      begin
         read(f, photo);
         write('   ', photo.nomFichier, '(', filepos(f), ') : ');
         i := 1;
         while (photo.tags[i] <> '') do
               begin
                  write(photo.tags[i], ', ');
                  i := i + 1;
               end;
          writeln;
      end;
end;


procedure RechercheTag (var f : FichierImages);
var
  t : string;
  photo : image;
  i, cmpt : integer;
begin
    reset(f);
    writeln('***Quel tag recherchez-vous :');
    readln(t);
    cmpt := 0;
    while (not (eof(f))) do
        begin
          read(f, photo);
          i := 1;

          // On parcours tous les tags, jusqu a trouver le tag ou avoir vu tous les tags
          while ((photo.tags[i] <> '') and (photo.tags[i] <> t)) do
                  i := i + 1;
          // Si le tag a ete trouve on l'affiche et on le comptabilise
          if  photo.tags[i] = t then
```

```
            begin
               writeln('--- Tag trouve dans photo ', photo.nomFichier);
               cmpt := cmpt + 1;
            end;

        end;
        // Affichage du nombre de photos contenant le tag recherche
        writeln('Nombre de photos trouvees : ', cmpt);
end;

function RecherchePhoto (var f : FichierImages) : integer;
var
  nom : string;
  photo : image;
begin
    reset(f);
    writeln('***Quel fichier recherchez-vous : ');
    readln(nom);
    photo.nomFichier := '';
    while (not (eof(f))) and (photo.nomFichier <> nom) do
       read(f, photo);
    if  photo.nomFichier = nom then
       begin
          writeln('--- Nom trouve -- position:', filepos(f)-1);
          RecherchePhoto := filepos(f)-1;
       end
    else
       RecherchePhoto := 0;
end;

procedure AjoutPhoto (var f : FichierImages);
begin
    seek(f, filesize(f));
    write(f, saisiePhoto);
end;

procedure AjoutTag (var f : FichierImages);
var
  i, j : integer;
  photo : image;
begin
    i := RecherchePhoto(f);
    seek(f, i);
    read(f, photo);
    writeln(' On va modifier la photo : ', photo.nomFichier);
    j := 1;
    while (photo.tags[j] <> '') do
      j := j+1;

    if j < N then
      begin
         writeln('   Quel tag voulez vous ajouter : ');
```

```
            readln(photo.tags[j]);
            photo.tags[j+1] := '';
            seek(f, i);
            write(f, photo);
        end
     else
        write(' Attention plus de place pour les tags');

end;

begin
   assign(fImages, 'MesPhotos.txt');
   //reset(fImages);
   rewrite(fImages);
   Repeat
      Repeat
         Writeln('================= Menu =================');
         Writeln('1-saisie Photos');
         Writeln('2-recherche tag');
         Writeln('3-ajout photo');
         Writeln('5-modifier photo / ajout tag');
         Writeln('6-afficher la liste des photos');
         Writeln('0-sortir du menu');
         writeln('========================================');

         Writeln('Entrer votre choix');
         Readln(choix);
      Until choix in [0..6];
      case choix of
         1:saisie(fImages);
         2:rechercheTag(fImages);
         3:ajoutPhoto(fImages);
//          4:suppression(fImages);
         5:AjoutTag(fImages);
         6:affichage(fImages);
      end;
   Until choix=0;
   close(fImages);
readln;
end.
```

{ ---------- pensez à changer de copie ---------- }

4ème Partie : Chaines de caractères *(5 points)*

On veut écrire un programme en Pascal qui permette d'extraire des chiffres présents dans un texte.

On considèrera qu'un texte est un ensemble de lignes.

1) Ecrire une procédure permettant la saisie du texte, et rangeant ce texte dans un tableau de lignes (avec 100 lignes au maximum).

   Chaque élément du tableau représentera une ligne de caractères, entrée au clavier par l'utilisateur, jusqu'au return (pour indiquer la fin de ligne). Cette procédure doit retourner le tableau ainsi composé.

2) Proposez une fonction qui, lors de son appel, reçoit une chaîne de caractères et retourne le nombre extrait de cette chaîne de caractères.

   Exemple : si la chaîne de caractères est "ser23df91" la fonction retourne le nombre entier 2391.

3) Ecrire une procédure à deux paramètres permettant de transformer le texte saisi précédemment (à la question 1) en une série de nombres, en utilisant la fonction de la question 2.

   Ces nombres seront rangés dans un tableau d'entiers dont chaque élément correspondra à une ligne du texte. Les deux paramètres sont le tableau de lignes et le tableau d'entiers.

4) Ecrire une fonction qui calcule la moyenne des nombres obtenus à la question 3. Cette fonction reçoit le tableau avec les nombres et retourne la moyenne.

5) Ecrire une procédure permettant d'afficher tous les nombres ainsi trouvés, pour le texte saisi par l'utilisateur, ligne par ligne ainsi que la moyenne calculée à la question 4.

```
program fin18;
type tableau_texte = array [1..100] of string[55] ;
   tableau_integer = array [1..100] of integer;

var
   tab:tableau_texte;
   tab_i:tableau_integer;
   T_r:integer;


procedure saisie_texte (var t:tableau_texte);
var    rep : char;
      ind : integer;
begin
      ind :=0;
      repeat
            ind:=ind+1;
            writeln('Entrez une chaine de caractères');
            readln(t[ind]);
            writeln('Voulez-vous continuer? o ou n');
            readln(rep);
      until (rep='n') or (ind=100);
      T_r:=ind;
end;

function str2num (e:string):integer;
var l_ch:integer;
   temp:integer;
   ind,nb:integer;
begin
      l_ch:=length(e);
      nb:=0;
      for ind:=1 to l_ch do
            begin
                  if e[ind] in ['0'..'9'] then
                              begin
```

```
                        temp:=ord(e[ind])-ord('0');
                        nb:=nb*10+temp;
                    end;
            end;
    str2num:=nb;
end;

procedure trans_texte(t:tableau_texte; var t_i:tableau_integer);
var ind:integer;
begin
    for ind:=1 to T_r do t_i[ind]:=str2num(t[ind]);
end;

procedure affichage (t:tableau_integer);
var
    ind:integer;
begin
    for ind :=1 to T_r do writeln(t[ind]);
end;

begin
    saisie_texte(tab);
    trans_texte(tab,tab_i);
    affichage(tab_i);
end.
```